

NAG Fortran Library Routine Document

F08QFF (STREXC/DTREXC)

Note: before using this routine, please read the Users' Note for your implementation to check the interpretation of *bold italicised* terms and other implementation-dependent details.

1 Purpose

F08QFF (STREXC/DTREXC) reorders the Schur factorization of a real general matrix.

2 Specification

```
SUBROUTINE F08QFF (COMPQ, N, T, LDT, Q, LDQ, IFST, ILST, WORK, INFO)
ENTRY          strex (COMPQ, N, T, LDT, Q, LDQ, IFST, ILST, WORK, INFO)
INTEGER       N, LDT, LDQ, IFST, ILST, INFO
real        T(LDT,*), Q(LDQ,*), WORK(*)
CHARACTER*1   COMPQ
```

The ENTRY statement enables the routine to be called by its LAPACK name.

3 Description

This routine reorders the Schur factorization of a real general matrix $A = QTQ^T$, so that the diagonal element or block of T with row index IFST is moved to row ILST.

The reordered Schur form \tilde{T} is computed by an orthogonal similarity transformation: $\tilde{T} = Z^T T Z$. Optionally the updated matrix \tilde{Q} of Schur vectors is computed as $\tilde{Q} = QZ$, giving $A = \tilde{Q}\tilde{T}\tilde{Q}^T$.

4 References

Golub G H and van Loan C F (1996) *Matrix Computations* (3rd Edition) Johns Hopkins University Press, Baltimore

5 Parameters

- 1: COMPQ – CHARACTER*1 *Input*
On entry: indicates whether the matrix Q of Schur vectors is to be updated, as follows:
 if COMPQ = 'V', the matrix Q of Schur vectors is updated;
 if COMPQ = 'N', no Schur vectors are updated.
Constraint: COMPQ = 'V' or 'N'.
- 2: N – INTEGER *Input*
On entry: n , the order of the matrix T .
Constraint: $N \geq 0$.
- 3: T(LDT,*) – **real** array *Input/Output*
Note: the second dimension of the array T must be at least $\max(1, N)$.
On entry: the n by n upper quasi-triangular matrix T in canonical Schur form, as returned by F08PEF (SHSEQR/DHSEQR).
On exit: T is overwritten by the updated matrix \tilde{T} . See also Section 8.

- 4: LDT – INTEGER *Input*
On entry: the first dimension of the array T as declared in the (sub)program from which F08QFF (STREXC/DTREXC) is called.
Constraint: $LDT \geq \max(1, N)$.
- 5: Q(LDQ,*) – *real* array *Input/Output*
Note: the second dimension of the array Q must be at least $\max(1, N)$ if COMPQ = 'V' and at least 1 if COMPQ = 'N' .
On entry: if COMPQ = 'V', Q must contain the n by n orthogonal matrix Q of Schur vectors.
On exit: if COMPQ = 'V', Q contains the updated matrix of Schur vectors.
 Q is not referenced if COMPQ = 'N'.
- 6: LDQ – INTEGER *Input*
On entry: the first dimension of the array Q as declared in the (sub)program from which F08QFF (STREXC/DTREXC) is called.
Constraints:
 $LDQ \geq \max(1, N)$ if COMPQ = 'V',
 $LDQ \geq 1$ if COMPQ = 'N'.
- 7: IFST – INTEGER *Input/Output*
 8: ILST – INTEGER *Input/Output*
On entry: IFST and ILST must specify the reordering of the diagonal elements or blocks of T . The element or block with row index IFST is moved to row ILST by a sequence of exchanges between adjacent elements or blocks.
On exit: if IFST pointed to the second row of a 2 by 2 block on entry, it is changed to point to the first row. ILST always points to the first row of the block in its final position (which may differ from its input value by ± 1).
Constraints:
 $1 \leq IFST \leq N$,
 $1 \leq ILST \leq N$.
- 9: WORK(*) – *real* array *Workspace*
Note: the dimension of the array WORK must be at least $\max(1, N)$.
- 10: INFO – INTEGER *Output*
On exit: INFO = 0 unless the routine detects an error (see Section 6).

6 Error Indicators and Warnings

INFO < 0

If INFO = $-i$, the i th parameter had an illegal value. An explanatory message is output, and execution of the program is terminated.

INFO = 1

Two adjacent diagonal elements or blocks could not be successfully exchanged. This error can only occur if the exchange involves at least one 2 by 2 block; it implies that the problem is very ill-conditioned, and that the eigenvalues of the two blocks are very close. On exit, T may have been partially reordered, and ILST points to the first row of the current position of the block being moved; Q (if requested) is updated consistently with T .

7 Accuracy

The computed matrix \tilde{T} is exactly similar to a matrix $T + E$, where

$$\|E\|_2 = O(\epsilon)\|T\|_2,$$

and ϵ is the *machine precision*.

Note that if a 2 by 2 diagonal block is involved in the re-ordering, its off-diagonal elements are in general changed; the diagonal elements and the eigenvalues of the block are unchanged unless the block is sufficiently ill-conditioned, in which case they may be noticeably altered. It is possible for a 2 by 2 block to break into two 1 by 1 blocks, that is, for a pair of complex eigenvalues to become purely real. The values of real eigenvalues however are never changed by the re-ordering.

8 Further Comments

The total number of floating-point operations is approximately $6nr$ if $\text{COMPQ} = \text{'N'}$, and $12nr$ if $\text{COMPQ} = \text{'V'}$, where $r = |\text{IFST} - \text{ILST}|$.

The input matrix T must be in canonical Schur form, as is the output matrix \tilde{T} . This has the following structure.

If all the computed eigenvalues are real, T is upper triangular and its diagonal elements are the eigenvalues.

If some of the computed eigenvalues form complex conjugate pairs, then T has 2 by 2 diagonal blocks. Each diagonal block has the form

$$\begin{pmatrix} t_{ii} & t_{i,i+1} \\ t_{i+1,i} & t_{i+1,i+1} \end{pmatrix} = \begin{pmatrix} \alpha & \beta \\ \gamma & \alpha \end{pmatrix}$$

where $\beta\gamma < 0$. The corresponding eigenvalues are $\alpha \pm \sqrt{\beta\gamma}$.

The complex analogue of this routine is F08QTF (CTREXC/ZTREXC).

9 Example

To reorder the Schur factorization of the matrix T so that the 2 by 2 block with row index 2 is moved to row 1, where

$$T = \begin{pmatrix} 0.80 & -0.11 & 0.01 & 0.03 \\ 0.00 & -0.10 & 0.25 & 0.35 \\ 0.00 & -0.65 & -0.10 & 0.20 \\ 0.00 & 0.00 & 0.00 & -0.10 \end{pmatrix}.$$

9.1 Program Text

Note: the listing of the example program presented below uses *bold italicised* terms to denote precision-dependent details. Please read the Users' Note for your implementation to check the interpretation of these terms. As explained in the Essential Introduction to this manual, the results produced may not be identical for all implementations.

```
*      F08QFF Example Program Text
*      Mark 16 Release. NAG Copyright 1992.
*      .. Parameters ..
      INTEGER          NIN, NOUT
      PARAMETER        (NIN=5,NOUT=6)
      INTEGER          NMAX, LDT, LDQ
      PARAMETER        (NMAX=8,LDT=NMAX,LDQ=1)
*      .. Local Scalars ..
      INTEGER          I, IFAIL, IFST, ILST, INFO, J, N
*      .. Local Arrays ..
      real            Q(LDQ,1), T(LDT,NMAX), WORK(NMAX)
*      .. External Subroutines ..
      EXTERNAL         strexc, X04CAF
*      .. Executable Statements ..
      WRITE (NOUT,*) 'F08QFF Example Program Results'
```

```

*      Skip heading in data file
      READ (NIN,*)
      READ (NIN,*) N
      IF (N.LE.NMAX) THEN
*
*          Read T from data file
*
      READ (NIN,*) ((T(I,J),J=1,N),I=1,N)
*
      READ (NIN,*) IFST, ILST
*
*      Reorder the Schur factorization T
*
      CALL strexc('No update',N,T,LDT,Q,LDQ,IFST,ILST,WORK,INFO)
*
*      Print reordered Schur form
*
      WRITE (NOUT,*)
      IFAIL = 0
*
      CALL X04CAF('General',' ',N,N,T,LDT,'Reordered Schur form',
+              IFAIL)
*
      END IF
      STOP
      END

```

9.2 Program Data

F08QFF Example Program Data

```

4      :Value of N
0.80  -0.11  0.01  0.03
0.00  -0.10  0.25  0.35
0.00  -0.65 -0.10  0.20
0.00   0.00  0.00 -0.10  :End of matrix T
2  1      :Values of IFST and ILST

```

9.3 Program Results

F08QFF Example Program Results

Reordered Schur form

	1	2	3	4
1	-0.1000	-0.6463	0.0874	0.2010
2	0.2514	-0.1000	0.0927	0.3505
3	0.0000	0.0000	0.8000	-0.0117
4	0.0000	0.0000	0.0000	-0.1000